Esimerkkinä Prologin käytöstä perinteisessä ohjelmoinnissa olkoon lisäys 2-3-puuhun [AHU83,§5.4;B01,§10.1].

2-3-puu on (haku)puu, jossa

• sisäsolmuilla on 2 tai 3 alipuuta, ja

• talletettu tieto on lehdissä. . .

• . . . jotka ovat samalla tasolla.

• Ohjelma haarautuu tapauksiin

parametrihahmojen mukaan.

• Kukin tapaus on oleellisesti vastaava

tasapaino(tus)ehto.

• Tietorakenneoperaatiot esitetään suoraan näillä parametrihahmoilla.

• Tietorakenteiden käsittely hahmonsovituksella on myös moderneissa funktionaalisissa kielissä kuten Haskell [T99] ja ML [CM98]. Niissä eri haarat ovat tietenkin toisensa poissulkevia, koska kullakin funktiokutsulla on vain yksi arvo.

(Listaus jatkuu yli kalvorajojen.)

/*

* e/0: Tyhjä 2-3-puu.

* l/1: Lehti jossa alkio (=luku).

* b/3: 2-haarainen sisäsolmu, välissä
       oikeanpuoleisen haaran pienin alkio.

* t/5: 3-haarainen sisäsolmu, väleissä
       oikeanpuoleisten haarojen pienimmät alkiot.

* c/3: Kuten b/3 mutta korkeus kasvanut yhdellä.
       Algoritmi pyrkii tasoittamaan syntyneen eron.

*/

% Pääpredikaatti:
% Kun luku X lisätään puuhun T0 niin saadaan puu T.
add23(T0,X,T):-
        ins23(T0,X,T1),
        put23(T1,T).

% Jos vielä lisäyksen lopussakin on korkeus kasvanut,
% niin koko puuta on kasvatettava uudella juurella:
put23(c(T0,X,TX),
      b(T0,X,TX)).
put23(l(X),
      l(X)).
put23(b(T0,X,TX),
      b(T0,X,TX)).
put23(t(T0,X,TX,Y,TY),
      t(T0,X,TX,Y,TY)).

% Tyhjään puuhun lisäys on lehti.
ins23(e,X,l(X)).

% Lehteen lisäys on korkeutta kasvanut taimi.
ins23(l(Y),X,c(l(Y),X,l(X))):-
        Y<X.
ins23(l(Y),X,c(l(X),Y,l(Y))):-
        X<Y.

% 2-haaraiseen sisäsolmuun lisäys:
% Valitse alipuu, lisää uusi alkio siihen, ja
% pistä kasvanut alipuu takaisin paikoilleen.
ins23(b(T0,Y,TY),X,T):-

        X<Y,
        ins23(T0,X,TX),
        put2l(TX,Y,TY,T).
ins23(b(T0,Y,TY),X,T):-
        Y<X,
        ins23(TY,X,TX),
        put2r(T0,Y,TX,T).

% 3-haaraiseen sisäsolmuun lisäys:
ins23(t(T0,Y,TY,Z,TZ),X,T):-
        X<Y,
        ins23(T0,X,TX),
        put3l(TX,Y,TY,Z,TZ,T).
ins23(t(T0,Y,TY,Z,TZ),X,T):-
        Y<X,
        X<Z,
        ins23(TY,X,TX),
        put3m(T0,Y,TX,Z,TZ,T).
ins23(t(T0,Y,TY,Z,TZ),X,T):-
        Z<X,
        ins23(TZ,X,TX),
        put3r(T0,Y,TY,Z,TX,T).

% Vaikka 2-haaraisen alipuu olisikin kasvanut korkeutta,
% ei hätää: tehdään sen tilalle 3-haarainen solmu!
put2l(c(T0,X,TX),Y,TY,
    t( T0,X,TX, Y,TY)).
put2l(b(T0,X,TX),Y,TY,
    b(b(T0,X,TX),Y,TY)).
put2l(t(T0,X,TX,Y,TY),Z,TZ,
    b(t(T0,X,TX,Y,TY),Z,TZ)).

put2r(T0,X,c(TX,Y,TY),
    t(T0,X, TX,Y,TY)).
put2r(T0,X,b(TX,Y,TY),
    b(T0,X,b(TX,Y,TY))).
put2r(T0,X,t(TX,Y,TY,Z,TZ),
    b(T0,X,t(TX,Y,TY,Z,TZ))).

% Jos 3-haaraisen alipuu on kasvanut korkeutta,
% on hätä: korjaus on tehtävä lähempänä juurta!
put3l(c(T0,X,TX),Y, TY,Z,TZ,
    c(b(T0,X,TX),Y,b(TY,Z,TZ))).
put3l(b(T0,X,TX),Y,TY,Z,ZY,
    t(b(T0,X,TX),Y,TY,Z,ZY)).
put3l(t(T0,W,TW,X,TX),Y,TY,Z,ZY,
    t(t(T0,W,TW,X,TX),Y,TY,Z,ZY)).

put3m(T0,X,c(TX, Y, TY),Z,TZ,

  c(b(T0,X, TX),Y,b(TY, Z,TZ))).
put3m(T0,X,b(TX,Y,TY),Z,TZ,
    t(T0,X,b(TX,Y,TY),Z,TZ)).
put3m(T0,W,t(TW,X,TX,Y,TY),Z,TZ,
    t(T0,W,t(TW,X,TX,Y,TY),Z,TZ)).

put3r(T0,X,TX, Y,c(TY,Z,TZ),
  c(b(T0,X,TX),Y,b(TY,Z,TZ))).
put3r(T0,X,TX,Y,b(TY,Z,TZ),
    t(T0,X,TX,Y,b(TY,Z,TZ))).
put3r(T0,W,TX,X,t(TX,Y,TY,Z,TZ),
    t(T0,W,TX,X,t(TX,Y,TY,Z,TZ))).

% Lisärutiineja puiden luomiseen ja katseluun:
build23([],e).
build23([X|Xs],T):-
        build23(Xs,T1),
        add23(T1,X,T).

show(T):-
        show(T,0).
show(e,_).
show(l(X),H):-
        view(X,H).
show(b(T0,X,TX),H):-
        H1 is H+5,
        view(-----,H),
        show(TX,H1),
        view(X,H),
        show(T0,H1),
        view(-----,H).
show(t(T0,X,TX,Y,TY),H):-
        H1 is H+5,
        view(-----,H),
        show(TY,H1),
        view(Y,H),
        show(TX,H1),
        view(X,H),
        show(T0,H1),
        view(-----,H).
view(A,H):-
        tab(H),
        write(A),
        nl.

(Listaus jatkui yli kalvorajojen.)

Esimerkkinä Prologin hyödyllisyydestä etsintäongelmissa olkoon lauselogiikan kaavojen toteutumisongelma:

• Syötteenä saadaan jakamattomista

atomikaavoista Pi konnektiiveilla ’ja’, ’tai’ ja ’ei’ yhdistelty kaava.

Esimerkiksi

(sataa P1 tai paistaa P2 ) muttei

eli ”ja ei”

sada P1 .

• Kysytään milloin se on totta.

Esimerkiksi silloin kun ei sada mutta paistaa.

Eli kun P1 on valetta mutta P2 totta.

• Riittää kirjoittaa induktiivinen totuusmääritelmä rekursiivisena Prolog-ohjelmana:

”Kaava ’F ja G’ on tosi jos sekä F että G on tosi”, jne.

• Kun atomikaavan P1 yhteen esiintymään laitetaan arvo tosi niin myös sen toinen esiintymä tulee kiinnitetyksi.

/*
onko(Kaava,Totuusarvo) on totta kun

1. Totuusarvo on joko ’tosi’ tai ’vale’.

2. Lauselogiikan Kaava on esitetty seuraavasti:

   a. Jakamattomat atomikaavat P_i on esitetty termeinä x(V_i)
      missä V_i on oma Prolog-muuttuja kullekin i.

   b. Konjunktiot ’F/\G’ eli ’F ja G’ on esitetty terminä ’ja(F,G)’.

   c. Disjunktiot ’F\/G’ eli ’F tai G’ on esitetty terminä ’tai(F,G)’.

   d. Negaatiot ’-F’ on esitetty terminä ’ei(F)’.

   Esimerkiksi kaava ’(A\/B)/\-A’ esitetään terminä
   ’ja(tai(x(A),x(B)),ei(x(A)))’.

3. Kaava saa Totuusarvon muuttujien V_i nykyisillä sidonnoilla.

*/

onko(x(T),T).

onko(ja(F,G),tosi):-
        onko(F,tosi),
        onko(G,tosi).
onko(ja(F,G),vale):-
        onko(tai(ei(F),ei(G)),tosi).

onko(tai(F,_),tosi):-
        onko(F,tosi).
onko(tai(_,G),tosi):-
        onko(G,tosi).
onko(tai(F,G),vale):-
        onko(ja(ei(F),ei(G)),tosi).

onko(ei(F),tosi):-
        onko(F,vale).
onko(ei(F),vale):-
        onko(F,tosi).

IV.4 Kirjastopalveluita

Tutustutaan Prolog-standardin peruspalveluihin listankäsittelyyn, aritmetiikkaan, samastukseen, negaatioon sekä syöttöön ja tulostukseen [B01,§2.7,3.1-3.2,3.4,5,6,7.2-7.3]. Toteutukset tarjoavat usein enemmänkin, varsinkin listoille.

IV.4.1 Listat

Myös Prolog tarjoaa kalvojen II.8 kaltaiset listat.

Prolog kirjoittaa listansa alkiot pilkuilla eroteltuina hakasulkeisiin: [tämä,on,listavakio].

Tyhjä lista on siis [].

Erillistä lainausmerkkimekanismia ei nyt tarvita, koska listan erottaa muista termeistä hakasulkeiden perusteella.

Listoja voi käsitellä kalvojen IV.5.1 listahahmoilla:

[A1,...,Ak|D]

on hahmo sellaiselle listalle, jossa on ainakin k alkiota.

Jos loppuhahmo ’|D’ puuttuu, on kyseessä hahmo tasan k-alkioiselle listalle (eli loppuhahmoksi otetaan ’|[]’).

Hahmossa kukin Ai ja D ovat Prolog-termejä. Listan i. alkiota yritetään samastaa termin Ai kanssa, ja loppulistaa termin D kanssa:

?- [tämä,on,lista]=[A1,A2|D].
A1 = tämä
A2 = on
D = [lista]

Näillä hahmoilla voidaan purkaa listoja kuten Scheme-kirjastofunktioilla cdk−1ar ja cdkr. Niillä voidaan myös rakentaa listoja lukemalla ne toisin päin: muodosta uusi lista lisäämällä listan D alkuun alkiot Ak, . . . , A1.

Esimerkiksi kahden listan liittäminen kolmanneksi on:

conc([],Y,Y).
conc([X|Xs],Y,[X|Z]):-
        conc(Xs,Y,Z).

• Jos ensimmäinen lista on tyhjä, niin toinen ja kolmas lista ovat samat.

• Jos ensimmäisellä listalla on ensimmäinen alkio X, niin kolmas lista alkaa tällä alkiolla ja jatkuu listana Z, joka puolestaan on ensimmäisen loppulistan Xs ja koko toisen listan Y liitos.

Tuloksella voi paitsi liittää myös jakaa listoja:

?- conc([1],[2],Z).

Z = [1, 2] ;

No
?- conc(X,Y,[1,2]).

X = []
Y = [1, 2] ;

X = [1]
Y = [2] ;

X = [1, 2]
Y = [] ;

No
?-

(Eikä ihme, sehän on kalvojen IV.2 p/3 listoille. . . )

IV.4.2 Aritmetiikka

Kalvoilla I.3 nähtiin Prolog-ohjelma (yksinkertaiseen) derivointiin.

Sen sääntöjen päissä käytettiin aritmeettisia operaattoreita +, *, / ja ** (potenssiin korotus) symbolisina kaksipaikkaisina funktoreina (jotka kirjoitettiin argumenttiensa väliin matematiikan tavalliseen tapaan).

Niiden avulla ohjelma haarautui ensimmäisen argumentin hahmon perusteella oikeaan sääntöön, ja tuotti toiseen argumenttiin vastaavan derivaattatermin vartalosta saatujen dervivaattatermien avulla.

Mutta ohjelman toisen säännön vartalossa eksponenttia N käytettiinkin numerona, josta vähennettiin 1, jotta tuloksena saataisiin uusi eksponentti N1.

Tämän sai aikaan (myös väliin kirjoitettu) kirjastopredikaatti V is E joka on totta kun (i) E on nykyisellä sidonnalla sellainen aritmeettinen lauseke jonka arvo voidaan laskea ja (ii) V sopii tuohon laskettuun arvoon.

Jos (i) ei pädekään, seuraa ajonaikainen virhe.

Kalvojen I.3 ohjelmassa kutsu N1 is N-1 on siis:

1. Tarkista, että lauseke N-1 on nykyisessä sidonnassa aritmeettinen:

• Ei vapaita muuttujia.

• Vain aritmeettisia funktoreita (kuten +/2, */2,

//2, **/2,. . . ).

• Vain numerovakioita.

Silloin lausekkeella on numeerinen arvo a.

Muuten kyseessä on ajonaikainen virhe.

2. Yritä sitoa N1 arvoon a.

Prolog-kielessä on äärellisen tarkkuuden kokonais- ja liukuluvut. Liian suurista kokonaisluvuista tulee automaattisesti liukulukuja, liian suurista liukuluvuista ääretön atomi $Infinity.

Myös (jälleen väliin kirjoitettavat) aritmeettiset vertailut =:=/2 (=), =\=/2 (≠), >/2 (>), >=/2 (≥),

</2 (<) ja =</2 (≤) laskevat argumenttiensa

numeroarvot samalla tavalla.

Maali ?-1<X,X<3. antaa siis virheen.

(Logiikkaohjelmointi voidaan tosin laajentaa tällaisiin reaalilukurajoitteisiin [B01,§14.2].)

IV.4.3 Samastus

Kalvoilla III.2 esiteltiin keskeinen samastusoperaatio

unify, jolla sääntöjärjestelmä teki annetuista

termeistä samoja sitomalla niiden muuttujia.

Samastusalgoritmin haara uvc eli ”vapaa muuttuja x vastaan rakenteinen termi t” käytti testiä

binding:occurs? varmistamaan, ettei x esiinny

termissä t. Muutenhan sidonnan x = t lisääminen aiheuttaisi kehän.

Tämä esiintymistarkistus (occurs check)

• hidastaa samastusta selvästi

• on harvoin tarpeen käytännön ohjelmoinnissa

joten Prologissa se jätetään pois – vaikka silloin kalvojen IV.5 puhdas rakenne likaantuu!

Puhtaan samastuksen voi tehdä kirjastopredikaatilla

unify with occurs check/2.

Myös Prologin omaa samastusta voi kutsua nimellä:

• =/2 luetaan ”samastuvat” (eli ”ovat samat”), jota käytettiin jo kalvojen IV.4.1 esimerkissä. (On myös ==/2 eli ”taysin samat ilman samastustakin”.)

• \=/2 luetaan ”eivät samastu”. (On myös \==/2.)

Esiintymistarkistuksen puuttuminen tekee näistä hieman arvaamattomia: ne saattavat esimerkiksi päätyä ikuiseen silmukkaan.

Prolog tarjoaa myös (jo kalvoilla IV.3 nähdyn) nimettömän muuttujan (eli pelkän alaviivan) joka samastuu aina mutta arvoa ei muistella. Esimerkiksi

f(a, , ) samastuu jokaiseen f(a,t1,t2) vaikka t1 ja

t2 eivät samastuisikaan keskenään.

Samastuksen kuluessa rakenteisia termejä puretaan osiinsa. Sama palvelu tarjotaan (väliin kirjoitettavalla) kirjastopredikaatilla =../2 (nimeltään ”univ”). Se on seuraava syntaktinen relaatio:

f(t1,...,tk) =.. [f,t1,...,tk]

(Myös k = 0 käy.) Sitä voidaan käyttää myös ”oikealta vasemmalle” eli kokoamaan termi annetusta listasta.

IV.4.4 Negaatiosta

Kalvojen IV.3 sukupuuesimerkissä olisimme halunneet kirjoittaa säännöt ”mies on terve (eli well) jos hän ei ole sairas (eli ill)”. Kuitenkin kirjoitimme nämä faktat erikseen.

Syynä on se, että Prolog-kielen negaatio (eli ”ei. . . ”) on jotakin muuta kuin odottaisi. Tähän palataan myöhemmin tutustuttuamme kielen merkitysoppiin tarkemmin.

Prolog-kielessä on negaatio not(p) missä p on kiellettävä alimaali. Sukupuuesimerkin sääntö olisi siis well(X) :- not(ill(X)).

(Voi kirjoittaa myös ilman sulkuja not ill(X).)

Tämä not/1 on eräänlainen korkeamman kertaluvun predikaatti, koska se ottaa argumentikseen toisen suoritettavan eli todistettavan faktan p. Prolog on kuitenkin ”filosofialtaan” ensimmäisen kertaluvun kieli, jossa on vain termejä ja niiden predikaatteja. Omien korkeamman kertaluvun laajennusten ohjelmointia olisi siis syytä välttää ”eleganssisyistä”.

Prolog-negaatio not(p) on negaatio epäonnistumisena (Negation as Failure): yritetään osoittaa p, ja jos tämä ei onnistu, niin silloin katsotaan, että on osoitettu ei-p.

Negaatio epäonnistumisena toimii kun p ei sisällä vapaita muuttujia sillä hetkellä kun kutsu not(p) tehdään.

Muuten tulos saattaa olla intuitiomme vastainen, joten tästä negaationkäyttöperiaatteesta kannattaa pitää kiinni! Siis: tuota ensin muuttujan X lopullinen arvo ja vasta sitten testaa sitä arvoa negatiivisella ehdolla not(q(X))!

Monissa Prolog-toteutuksissa negaation nimi onkin \+ korkostamaan sitä, ettei kyseesä olekaan ”oikea” (eli kielellisen intuitiomme mukainen) negaatio.

IV.4.5 Haun ohjailusta

Kalvojen IV.4.4 negaatio epäonnistumisena voitaisiin ohjelmoida predikaattina

not(P) :- call(P),!,fail.
not(_).

• call(P) tulkitsee muuttujan P arvon faktana, ja tekee vastaavan aliohjelmakutsun.

Joissakin toteutuksissa riittääkin kirjoittaa suoraan P.

Vertaa Scheme-kirjastofunktio eval kalvoilla II.9.2.

• Huutomerkki ’!’ eli katkaisupredikaatti (Cut) ohjaa Prolog-ohjelman suoritusta eli todistuksen hakua.

Kalvojen IV.3 aliohjelmatulkinnassa se voidaan lukea ”jos suoritus peruuttaa taaksepäin yli aliohjelmassa nimeltä q esiintyvän huutomerkin, niin suoritus putoaa suoraan ulos aliohjelman q alareunasta yrittämättä sen muita vaihtoehtoja”.

• Nyt huutomerkki sanoo siis ”P onnistui yhdellä tavalla, älä turhaan etsi muita tapoja, tämä tapa kelpaa minulle”.

Huutomerkillä voi siis parantaa tehokkuutta kertomalla järjestelmälle, että nyt on löydetty se tulos, joka haluttiin — etsintä voidaan tältä osin lopettaa. Tähän palataan myöhemmin.

• fail epäonnistuu aina (eli sillä ei ole

määritelmää).

Negaation määritelmän lisäksi fail on hyödyllinen esimerkiksi silloin, kun Prolog-ohjelman täytyy tehdä sivuvaikutuksenaan jotakin. Tähän palataan myöhemmin tietokantojen yhteydessä.

Siis not-predikaattimme luetaan seuraavasti:

”Jos aliohjelmakutsu P onnistuu, niin rajoita not(P) siihen ratkaisuun, jota ei ole.

Muuten not(P) onnistuu.”

Siispä se on ”negation as failure”.

IV.4.6 Syöte ja tulostus

Prolog-kielen syöte ja tulostus noudattaa samaa ajattelutapaa kuin Scheme-kielessäkin (kalvot II.9.7–II.9.8):

• Syöte ja tulostus voidaan ohjata portteihin [joita kutsutaan virroiksi (streams)]

• Voidaan lukea (read/1 yms.) ja kirjoittaa (write/1 yms.) kokonaisia Prolog-termejä.

Yksityiskohdat sivuutetaan. (Toteutukset eivät aina tue kaikkia standardin piirteitä.)

Periaatteessa Prologin pitäisi pystyä peruuttamaan myös syötteenluku- ja tuloksenkirjoitusoperaatiot, mutta tämä olisi käytännössä liian vaikeaa: luettu syöte pitäisi palauttaa takaisin ja kirjoitettu tuloste pyyhkiä pois.

Näin ei siis tehdä, joten syötteen ja tulostuksen oikea käyttö vaatii Prolog-suoritusjärjestyksen tuntemista.

Esimerkiksi kutsu read(X) lukee muuttujaan X (nykyisestä syöteportista) seuraavan Prolog-termin t ja sen perässä olevan pisteen ’.’.

Peruutettaessa ei löydy muita ratkaisuja ja t (pisteineen) pysyy luettuna.

Jos X onkin termi u, niin t luetaan (pisteineen) ensin, ja sitten yritetään samastaa t ja u. Jos samastus onnistuu, niin kutsukin onnistuu; muuten kutsu epäonnistuu – mutta molemmissa tapauksissa t on luettu (pisteineen)!

Lukea (ja kirjoittaa) voi myös sääntöjä: ne tulkitaan termeiksi ’:-’(pää,vartalo) missä

• funktori :-/2 kirjoitetaan väliin

• vartalo on joko yksinäinen termi tai sitten termi,vartalo

ja muuttujat numeroidaan uudelleen.

Prolog-ohjelman lataus(kirjasto)predikaatti

consult(’tiedostonimi’) (vertaa Scheme-kalvoja

II.9.9) lukeekin readilla (vertaa Scheme-kalvoja II.9.8) termejä, ja liittää ne nykyisen ohjelman loppuun.

IV.5 Semantiikka

Logiikkaohjelmoinnissa ohjelmien merkitystä voidaan kuvailla kolmesta eri näkökulmasta:

Suorituslähtöisesti muiden ohjelmointikielten tapaan [B01,§2.4].

”Mitä askeleita ohjelma tekee?”

Kalvojen IV.2 ohjelma sum siirtää toisen argumenttinsa kolmanteen, ja sen päälle kaikki ensimmäisen s-funktorit.

Näin tehtiin kalvoilla IV.3. Kalvojen III tulkki on tämän näkökulman toteutus.

Malliteoreettisesti käyttäen loogisen totuuden käsitettä [B01,§2.3;L87,§6].

”Millaisen maailman ohjelma kuvailee?”

Ohjelma sum kertoo mitä merkitsee väite ”kolmas luku on kahden ensimmäisen summa”.

(Loogiset yksityiskohdat sivuutetaan; logiikkaa opiskelleet kohtaavat tuttujen käsitteiden erikoistapauksia.)

Todistusteoreettisesti käyttäen loogisen todistuksen käsitettä [L87,§7-10].

Ohjelma sum on aksiomatisointi jolla voidaan todistaa näitä väitteitä ”kolmas luku on kahden ensimmäisen summa”.

• Malliteoria kuvaa ohjelman merkityksen ilman viittauksia mihinkään laskentasääntöihin.

(Ohjelma on siis symbolikasauma, jolle ihmisellä on mielessään tulkinta. . . )

• Todistusteoria kuvaa sellaiset mahdolliset laskentasäännöt, jotka ovat malliteorian mukaisia.

(. . . joten säännöt ovat siis niitä symbolisia operaatioita, jotka eivät sodi tulkintaa vastaan. . . )

• Suorituskone toteuttaa (jonkin) strategian jolla todistusteorian sallimia laskentasääntöjä yritetään soveltaa laskentatehtäviä ratkaistaessa.

(. . . joten ohjelmoidaan tietokone ratkomaan laskutehtäviä näillä operaatioilla!)

Samaa kolminaisuutta käytetään yleisemminkin ohjelmointikielten teoriassa, mutta logiikkaohjelmoinnissa se näkyy harvinaisen puhtaana.

IV.5.1 Suoritussemantiikka

Tarkastellaan predikaatin p/k määritelmää n säännöllä: p(t(1,1), . . . , t(1,k)) :- q(1,1), . . . , q(1,m_1). . p(t(n,1), . . . , t(n,k)) :- q(n,1), . . . , q(n,m_n). Sen voi lukea määritelmänä aliohjelmalle nimeltä p jolla on k parametria.

Jokaisessa haarassa 1 ≤ i ≤ n termit t(i,1), . . . , t(i,k) määräävät hahmon johon aliohjelmakutsun p(τ1, . . . , τk) argumenttien τ1, . . . , τk on sovittava (kalvojen III.2) samastuksen mielessä.

Haara i on mahdollinen jos argumentit sopivat hahmoon; muuten haara sivuutetaan. Hahmo siis kertoo minkä ”muotoisille” syötteille τj tämä haara on tarkoitettu.

Koska nyt samastetaan argumentteja parametrihahmoihin, on mukana myös sidonta θ jota laajennetaan suorituksen edetessä, ja joka hoitaa parametrinvälityksen. (Vrt. kalvojen II.9.1 ympäristöt.)

Loogikot kirjoittavat τθ operaatiomme

(binding:expand τ θ) vastineen.

(Ali)ohjelman lukuperiaate on tuttu ”ylhäältä alas, vasemmalta oikealle”:

Etsitään ensimmäinen mahdollinen haara i, eli sellainen jolla operaation θ0 = (unify p(τ1, . . . , τk) p(t(i,1), . . . , t(i,k)) θ) vastine onnistuu.

Ensin kuitenkin nimettiin haaran i muuttujat uudelleen, jotta vanhat muuttujanarvot θ ja uudet parametrit eivät sekoittuisi keskenään.

Jatketaan kokeiltavan haaran i vartaloon, eli

1. tehdään samalla tavalla kutsu q(i,1) sidonnalla θ0, josta saadaan vastaukseksi sidonta θ1,

2. tehdään kutsu q(i,2) sidonnalla θ1, josta saadaan sidonta θ2,. . .

Lopulta vastataan kysyjälle että kutsu p(τ1, . . . , τk) laajensi alkusidonnat θ loppusidonnoiksi θm_i.

Entä jos jokin seuraava kutsu q(i,j+1) ilmoittaakin, ettei se pysty laajentamaan edelliseltä kutsulta q(i,j) saamaansa sidontaa θj sidonnaksi θj+1?

Silloin peruutetaan (backtrack) takaisin edelliseen kutsuun q(i,j), ja pyydetään siltä jotakin toista sidonnan θj−1 laajennusta sidonnaksi θ′j.

Jos tämän kokeiltavan haaran i vartalosta peruutetaan takaisin päähän asti, niin tämä kutsu p(τ1, . . . , τk) sidonnoilla θ on sittenkin suoritettava käyttäen jotakin toista haaraa.

Jatketaan siis seuraavan mahdollisen haaran etsimistä haarasta i + 1 alkaen. Jos sellainen löytyy, niin kokeillaan sitä.

Jos taas haarat loppuvat kesken, niin vastataan kysyjälle ”kutsu p(τ1, . . . , τk) ei pysty laajentamaan sidontaa θ”. Tämä aiheuttaa peruutusta vuorostaan kysyjässä.

Vastaavasti myös kysyjältä voi tulla pyyntö ”vaikka kutsu p(τ1, . . . , τk) laajensikin sidontansa θ onnistuneesti sidonnaksi θm_i haaralla i, ei tulos sopinutkaan myöhempään laskentaan, joten voisinko sittenkin saada jonkin toisen vaihtoehdon θ′m_i?”

Silloin peruutetaan takaisin viimeiseen onnistuneeseen kutsuun q(i,m_i), ja yritetään laajentaa sen saamaa sidontaa θm_i−1.

Perimmäisenä kysyjänä on käyttäjä, joka saattaa olla tyytymätön hänelle tulostettuun vastaukseen.

Prolog-ohjelmien suorituksessa vuorottelee siis:

Etenemisvaihde jonka aikana suoritetaan aina heti seuraava mahdollinen haara.

Kun näin ajaudutaan umpikujaan, kytketään päälle. . .

Peruutusvaihde jonka aikana ohjelman suoritusta ”kelataan takaisinpäin” edelliseen sellaiseen kohtaan, joka olisi voitu tehdä toisinkin.

Siinä kohdassa valitaankin nyt toisin, ja jatketaan taas etenemisvaihteella. . .

Samoin aliohjelmilla ei ole kiinteitä syöte- ja tulosparametreja, vaan jokainen kutsu lukee syötteenään sille annettua sidontaa, ja tuottaa tuloksenaan siitä laajennettua versiota.

Esimerkiksi predikaattia p/3 voi käyttää laskemaan yhteen 1 + 2 kutsulla p(s(o),s(s(o)),Z) eikä mitään peruutusta tapahdu, koska etenemisvaihtoehtoja on aina tasan yksi, ja lopuksi onnistutaan saamaan vastaus Z=s(s(s(o))).

Kääntäen luku 3 voidaan hajoittaa summiksi 0 + 3, 1 + 2, 2 + 1, 3 + 0 kutsulla p(X,Y,s(s(s(o)))) jossa kukin käyttäjän pyytämä peruutus tuottaa seuraavan vastauksen.

pajakari$ /opt/pl/bin/pl
Welcome to SWI-Prolog (Version 4.0.1)
Copyright (c) 1990-2000 University of Amsterdam.
Copy policy: GPL-2 (see www.gnu.org)

For help, use ?- help(Topic). or ?- apropos(Word).

?- consult(’sum.pl’).
% sum.pl compiled 0.01 sec, 652 bytes

Yes
?- p(s(o),s(s(o)),Z).

Z = s(s(s(o))) ;

No
?- p(X,Y,s(s(s(o)))).

X = o
Y = s(s(s(o))) ;

X = s(o)
Y = s(s(o)) ;

X = s(s(o))
Y = s(o) ;

X = s(s(s(o)))
Y = o ;

No
?-

IV.5.2 Mallisemantiikka

Olkoon L tarkasteltava ohjelma. Sen sisältämät funktorit voidaan (yleensä) jakaa predikaatteihin PL ja muihin funktoreihin FL.

Esimerkiksi Psum = {p/3} ja Fsum = {o/0, s/1}.

(Ranskalainen Jacques Herbrand kehitti todistusteoriaa 1930-luvulla.)

Annetun ohjelman L Herbrandin. . .

Universumi UL on ne muuttujattomat termit jotka voidaan muodostaa ilman predikaatteja, eli pienin sellainen joukko, että

• jos a/0 ∈ FL, niin myös a/0 ∈ UL ja • jos f/k ∈ FL ja τ1, . . . , τk ∈ UL (missä k > 0), niin myös f(τ1, . . . , τk) ∈ UL. Usum = {o} ∪ {s(o)} ∪ {s(o), s(s(o))} . = {sno : n ∈ N } .

Kanta BL on ne faktat jotka voidaan muodostaa universumin UL termeistä, eli BL = {p(τ1, . . . , τk) : p/k ∈ PL} . Bsum = {

p(sao,sbo,sco): a, b, c ∈ N

} .

Malli on mikä tahansa kannan BL osajoukko, johon on poimittu tosiksi ainakin niin paljon faktoja, että ohjelman L kaikki lauseet (clauses, eli säännöt ja faktat) ovat tosia:

Olkoon θ mielivaltainen muuttujien sidonta universumin UL alkioihin.

• Malliin poimitaan (muuttujaton) fakta pθ ohjelman L jokaiselle faktalle p.

Haluamme näet pitää totena faktaa p kaikilla mahdollisilla sen muuttujien saamilla arvoilla.

• Jos malliin on jo poimittu faktat q1θ, . . . , qnθ ja ohjelmassa on sääntö p:-q1,. . . ,qn. niin poimitaan myös pθ.

Haluamme näet pitää totena ehtolausetta

jos q1 ja . . . ja qn niin myös p

jonka loogikko kirjoittaisi implikaationa ∀ ⃗X.q1 ∧ . . . ∧ qn → p missä ⃗X kattaa kaikki säännön muuttujat. (Ns. Hornin lauseena.)

Osoittautuu että on olemassa pienin malli ML ⊆ BL, ja se saadaan poimimalla mukaan vain ne faktat, jotka nämä ehdot pakottavat, eikä mitään muuta.

Tämä ML on ohjelman L deklaratiivinen merkitys, ja ohjelman L suorittaminen annetulla (muuttujattomalla) maalifaktalla p on kysymykseen ”Onko p ∈ ML?” vastaamista.

(Tarkkaan ottaen joukkoon FL on lisättävä myös maalifaktan p funktorit. . . )

Kalvojen IV.5.1 menetelmä on yksi algoritmi vastauksen laskemiseksi, kuitenkin epätäydellinen:

• algoritmi saattaa joutua ikuiseen silmukkaan löytämättä malliteorian mukaista vastausta

• koska algoritmi kokeilee sääntöjä kiinteässä järjestyksessä, joka saattaa olla tällä kertaa väärä. Malliteoriassa säännöt ovat tasa-arvoisia, mutta sellainen algoritmi söisi liikaa työmuistia.

Faktasta: p(o,sbo,sbo) ∈ Msum kaikilla b ∈ N .

Säännöstä: jos p(sao,sbo,sco) ∈ Msum niin myös

p(sa+1o,sbo,sc+1o) ∈ Msum.

Minimaalisuudesta: muuta ei mallissa Msum ole.

Osoitetaan Msum = {

p(sao,sbo,sa+bo): a, b ∈ N

} :

”⊇” : Olkoon p(sao,sbo,sa+bo) mielivaltainen.

Näytetään induktiolla että ∈ Msum: a = 0 : Seuraa suoraan faktasta. a > 0 : Seuraa induktio-oletuksesta

p(sa−1o,sbo,sa−1+bo) ∈ Msum,

ja siihen sovelletusta säännöstä.

”⊆” : Olkoon ρ = p(sao,sbo,sco) ∈ Msum mielivaltainen. Näytetään induktiolla suureen a suhteen, että c = a + b.

Miten ρ ilmestyi joukkoon Msum?

• Jos käytettiin faktaa, niin a = 0 ja b = c, ja siis c = a + b.

• Jos käytettiin sääntöä, niin sen vartalolle

p(sa−1o,sbo,sc−1o) ∈ Msum,

ja siitä induktio-oletuksella c − 1 = a − 1 + b eli c = a + b.

Minimaalisuuden nojalla muita tapoja ei ole.

Kirjallisuutta

AS96 H.Abelson&G.J.Sussman: Structure and Interpretation of Computer Programs. Second Edition. MIT Press, 1996.

AHV95 S.Abiteboul,R.Hull,V.Vianu: Foundations of Databases. Addison Wesley, 1995.

AHU83 A.V.Aho,J.E.Hopcroft&J.D.Ullman: Data Structures and Algorithms. Addison Wesley, 1983.

B01 I.Bratko: Prolog Programming for Artificial Intelligence. Third Edition. Addison Wesley, 2001.

CM87 W.F.Clocksin&C.S.Mellish: Programming in Prolog. Third, Revised and Extended Edition. Springer-Verlag, 1987.

CM98 G.Cousineau&M.Mauny: The Functional Approach to Programming. Cambridge University Press, 1998.

DD98 C.J.Date&H.Darwen: Foundation for Object/Relational Databases: The Third Manifesto. Addison Wesley, 1998.

DEC96 P.Deransart,A.Ed-Dbali&L.Cervoni: Prolog:

The Standard. Reference Manual. Springer-Verlag, 1996.

D96 R.K.Dybvig: The Scheme Programming Language: ANSI Scheme. Second Edition. Prentice Hall, 1996.

H92 D.Harel: Algorithmics. Second Edition. Addison Wesley, 1992.

H97 J.R.Hindley: Basic Simple Type Theory.

Cambridge University Press, 1997.

KCR98 R.Kelsey, W.Clinger&J.Rees (toim.): Revised5 Report on the Algorithmic Language Scheme. Higher-Order and Symbolic Computation 11(1), 1998 tai ACM SIGPLAN Notices 33(9), 1998.

L87 J.W.Lloyd: Foundations of Logic

Programming. Second, Extended Edition. Springer-Verlag, 1987.

P02 B.C.Pierce: Type Theory and Programming Languages. MIT Press, 2002.

S98 S.Slade: Object-Oriented Common Lisp.

Prentice Hall, 1998.

SS86 L.Sterling&E.Shapiro: The Art of Prolog. MIT Press, 1986.

S97 B.Stroustrup. The C++ Programming Language. Third Edition. Addison-Wesley, 1997.

T99 S.Thompson: Haskell: The Craft of Functional Programming. Second Edition. Addison-Wesley, 1999.